Capítulo 9

Séries infinitas; número *e* de Euler e o número Pi

A série de Seno

A seguinte fórmula representa um polinômio:

$$S_n = a_0 + a_1 x + a_2 x^2 + ... + a_n x^n \quad (1)$$

Os termos das séries de potências são funções de x da forma $b_i = a_i x^i$. Os polinômios (1) são as somas parciais das séries de potências. Funções como e^x, sen(x), cos(x) e outras podem ser escritas na forma dum desenvolvimento em uma série de potências

$$f(x) = a_0 + a_1 x + a_2 x^2 + ... \quad (2)$$

Como primeiro exemplo consideramos o seguinte desenvolvimento para a função sen(x), que será válido para todos os x

$$sen(x) = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + -... \quad (3)$$

sendo x o ângulo em radianos. O símbolo ! é a fatorial, ou seja n! = 1·2·3···n é o produto dos primeiros n inteiros. Os termos $x^n/n!$ tendem a 0 à medida que n tende para o infinito. Uma vez que os termos da série são de sinais alternados e decrescentes, decorre que o erro cometido quando somamos somente um certo número n de termos, n arbitrário, por exemplo 3, não excederá o valor do primeiro termo desprezado.

Exemplo: Qual é o seno de 1° (= π/180 radianos)?
Da série (3) decorre

$$\text{sen}(\pi/180) = \pi/180 - (\pi/180)^3/6 + ... = 0{,}0174524064...$$

O primeiro termo desprezado é $(\pi/180)^5/120 = 1{,}3496...10^{-11}$, que é menor do que 0,000 000 000 02. O erro cometido em tomar só dois termos não é maior do que $2 \cdot 10^{-11}$, portanto sen (1°) = 0,0174524064, com dez casas decimais.

Não é difícil implementar (3) numa planilha:

D3 =C3^A3/FATORIAL(A3)

	A	B	C	D	E	F	G	H	I	J
1	1	1	0,01745329	0,017453293			x=	0,017453	(=1 grau)	
2	3	-1	-0,0174533	-8,86096E-07						
3	5	1	0,01745329	1,3496E-11		sen(x)=				
4						0,0174524064				
5		**Série de Seno**								
6										
7										
8										

A série (3) não é útil para valores de x > 20 radianos. Por exemplo, para calcular sen(20), é preciso somar 40 termos da série (3) para obter um resultado correto com oito casas decimais (= 0,91294525...).
Por outro lado, não precisamos calcular com valores de x > 20, pois sempre podemos restar múltiplos de 2π até obter um valor aceitável.
(20 - 3·2π = 1,150444078 e sen(1,150444078) = 0,9129452507. Para obter sen(1,150444078) com 8 casas decimais, devemos somar 20 termos da série (3)).

As entradas para a planilha são:

A1: 1
A2: =A1+2, copiar para baixo
B1: 1
B2: =B1*(-1), copiar
C1: =B1*H$1, copiar; D1: =H$1
D2: =C2^A2/FATORIAL(A2), copiar

Gostaria mencionar uma **técnica recursiva** para o cálculo da série (3).
Se tomarmos y1:= x, então o segundo termo será y2 = $-x^2/(2\cdot3)\cdot$y1. Do segundo termo obtemos y3 = $-x^2/(4\cdot5)\cdot$y2 ...
Este processo podemos escrever da seguinte maneira

$$y_1 = x,\ \ y_{k+1} = -\frac{x^2}{2k(2k+1)},\ \ k = 1,2,\dots,n-1 \quad (4)$$

B6 =(-1)*E$1^2/(2*A5*(2*A5+1))*B5

	A	B	C	D	E	F	G	H	I	J
1				x=	2					
2										
3										
4	k	y	soma		sen(x)=	0,90929743		**Seno recursivamente**		
5	1	2	2							
6	2	-1,33333	0,666667							
7	3	0,266667	0,933333							
8	4	-0,0254	0,907937							
9	5	0,001411	0,909347							
10	6	-5,1E-05	0,909296							

As entradas são

A5: 1 (=k); B5: =E$1 (=y1)
Na coluna C temos a soma C5: =B5
A6: =A5+1
B6: =(-1)*E$1^2/(2*A5*(2*A5+1))*B5
C6: =C5+B6
Copiar A6:C6 até A6:C104 (soma-se 100 termos); o resultado fica na célula
E5: =C104
Para x = 20 obtemos a soma 0,9129452534737. Este número é correto com 8 casas decimais.

A fórmula de recursão (4) pode ser escrita como Sub-rotina ou como Função:

```
Sub seno()
    Dim n As Integer, k As Integer
    Dim fator As Double, s As Double
    Dim x As Double, y As Double
    x = InputBox("ângulo (radianos)?")
    y = x: n = 100: s = x
    For k = 1 To n - 1 Step 1
        fator = 1 / (4 * k ^ 2 + 2 * k)
        y = (-1) * fator * x * x * y
        s = s + y
    Next

    MsgBox "sen(" & x & ") = " & Format(s, "0.0000")

 End Sub
```

```
Function Sinus(x As Double, n As Integer) As Double
    Dim k As Integer
    Dim fator As Double, s As Double
    Dim y As Double

    y = x:  s = x
    For k = 1 To n - 1 Step 1
        fator = 1 / (4 * k ^ 2 + 2 * k)
        y = (-1) * fator * x * x * y
        s = s + y
    Next
    Sinus = s

End Function
```

Observe como definimos o formato do resultado usando a função **Format**:

MsgBox "sen(" & x & ") = " & Format(s, "0.0000")

Imagino que você esta interessado em fazer alguns **exemplos adicionais** para praticar o VBA. Trate de implementar as seguintes séries de potências:

$$\ln(1+x) = x - \frac{x^2}{2} + \frac{x^3}{3} - \ldots, \text{ válido para } -1 < x \leq 1 \quad (5)$$

$$\ln(z) = 2(x + \frac{x^3}{3} + \frac{x^5}{5} + \ldots), \text{ com } x = \frac{z-1}{z+1}, \text{ z} > 0 \quad (6)$$

$$e^x = 1 + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} + \ldots, \text{ válido para todos os } x \quad (7)$$

Dicas:

Para (6): t=x, s=x, k=1

Depois: k = k+2 e t=x·x·t; s=s+t/k. (t = variável temporária)
controle: ln(40) = 3,688876... , correto com 3 casas decimais

Para (7): y1=1, y2=x·y1/1, y3=x·y2/2 etc.

Resolução

Para (6) e (7):

E2: =(E$1-1)/(E$1+1)
A5: 1 (=k); B5: =E$2 (=variável auxiliar t)
C5: =E$2 (= termo da soma)
A6: =A5+2
B6: =E$2*E$2*B5 (= potências de x)
C6: =C5+B6/A6 (=soma parcial)

Copiar A6:C6 até A6:C104 (F5, Ctrl+d), soma-se 100 termos. A soma dos 100 termos deve ser multiplicada por 2 e fica em E5: =2*C104



H4 =logz(50)

	A	B	C	D	E	F	G	H	I
1				z=	50				
2				x=	0,960784				
3								Function logz	
4				ln(z)=	3,911985		logz=	3,911985	
5	1	0,960784	0,960784						
6	3	0,886906	1,25642						
7	5	0,818709	1,420162						
8	7	0,755756	1,528127						
9	9	0,697643	1,605642						
10	11	0,643999	1,664188						

A seguinte função "logz" calcula ln(z) somando 100 termos da série (6). Com o termo auxiliar t evitamos o cálculo direto das potências de x.

A função "Expo" calcula a função exponencial e^x somando n+1 termos da série (7). Para x = 1 obteremos para n = 10 o valor e^1 = **e** = 2,7182818. É muito interessante, estudar com "Expo" a dependência do valor de *e* do número de termos somados.

```
Function logz(z As Double) As Double
    Dim x As Double, s As Double, t As Double
    Dim k As Integer

    x = (z - 1) / (z + 1)
    s = x: t = x ' t=variável auxiliar
    For k = 3 To 199 Step 2
        t = x * x * t ' =x^k
        s = s + t / k
    Next
    logz = 2 * s

  End Function
```

```
Function Expo(x As Double, n As Integer) As Double
    Dim y As Double, s As Double, k As Integer

    y = 1
    For k = 1 To n Step 1
        y = x * y / k
        s = s + y
    Next
      Expo = s + 1

End Function
```

O número e de Euler (1707-1783) e o método de Horner

Já no capítulo três, falando de juros contínuos, mencionamos o número de Euler: **e** = 2,71828... Até a época de Euler, a função exponencial e^x era considerada meramente como o inverso da função logarítmica. Euler colocou as duas funções numa base igual, dando-lhes definições independentes:

$$e^x = \lim_{n\to\infty}(1+\frac{x}{n})^n$$
$$\ln x = \lim_{n\to\infty} n(x^{\frac{1}{n}}-1) \qquad (8)$$

Euler já tinha usado a letra *e* para representar o número 2,71828 ... num de seus primeiros trabalhos, um manuscrito intitulado "*Meditação sobre Experimentos feitos recentemente sobre o disparo do Canhão* ", escrito em 1727.

Como vimos, podemos expressar estes limites em forma de séries infinitas. A série (7) permite o cálculo de **e** com qualquer grau de precisão desejado. Para calcular *e* efetivamente, não é preciso calcular os fatoriais, pois, como já mostramos nas macros, pode-se escrever a série para *e* na forma

$$e = 1+(1+1/2(1+1/3(1+1/4(1+1/5(1+1/6(1+...)))))) \quad (9)$$

É recomendável calcular esta expressão desde o interior até o lado exterior:

n	soma
5	1+1/5=1,2
4	1+(1,2)/4 = 1,3
3	1+(1,3)/3 = 1,43333
2	1+(1,43333)/2 = 1,71666
1	1+1,71666 = 2,71666

Para realizar este esquema, precisamos de só duas colunas


B6 =1+B5/A6

	A	B	C	D	E	F
1						
2		Número de Euler				
3						
4						
5	15	1,066667				
6	14	1,07619		e=	2,718281828459	
7	13	1,082784				
8	12	1,090232				
9	11	1,099112				
10	10	1,109911				
11	9	1,123323				
12	8	1,140415				
13	7	1,162916				
14	6	1,193819				
15	5	1,238764				
16	4	1,309691				
17	3	1,436564				
18	2	1,718282				
19	1	2,718282				
20						

A5: 15 (=número dos termos)
B5: =1+1/A5; A6: =A5-1; B6: =1+B5/A6

Copiar A6:B6 até A6:B19. Em E6: =B19 temos o valor de **e** , correto com 12 casas decimais.

Um método parecido a este é o método de **Horner**, que se utiliza para o cálculo de valores de polinômios como (1). Este algoritmo torna-se computacionalmente muito eficiente, pois necessita de um total de n operações de adição e n operações de multiplicação.

Exemplo: Seja $P(x) = 4x^3 - 2x^2 + 3x - 6$. Determine, utilizando o método de multiplicação aninhada de Horner, P(3).

Solução: Escrevemos o polinômio na seguinte forma aninhada:

$$P(x) = ((4x-2)x+3)x-6$$

Observamos a regra **P = P·x + A(i)** com A(3) = -6, A(2) = 3, A(1) = -2 e A(0) = 4. Os A(i) são os coeficientes dos termos do polinômio dado. Colocando o valor x = 3 na forma segundo Horner, obteremos

$$P(3) = ((4\cdot 3-2)3+3)3 - 6 = 93$$

Para implementar a regra P = P·x + A(i) numa planilha de Excel, escrevemos os coeficientes A(i) na coluna A e a regra uma só vez na célula B6 (depois copiar até for necessário). O último valor na coluna B é o valor de P(x) para o x dado.

B6 | f_x | =B5*F$1+A6

	A	B	C	D	E	F	G
1					x=	3	
2			**Horner**				
3							
4	A(i)	P=P*x+A(i)					
5		0					
6	4	4					
7	-2	10					
8	3	33					
9	-6	93					
10							

A Sub-rotina "Horner_sub" demonstra, como se pode trabalhar com uma **lista** de números dentro de um **vetor** a(i). Os a(i) são valores indexados. Lembre-se que A(0) é o primeiro coeficiente do polinômio: $P(x) = A(0)x^n + A(1)\ x^{n-1} + ...+A(n-1)x+A(n)$.

Exemplo: Seja $P(x) = 3x^4 - 6x^3 - 2x^2 + 5x - 8$.   x =3

A macro pergunta, primeiro, o valor de x, depois o grau (n = 4). Em seguida, ela pede o valor de a(0) (= 3) etc. até a(4) (= -8)

A MsgBox dá sucessivamente todos os coeficientes do polinômio. (Pode-se obter, também, uma lista dos coeficientes.)

Finalmente aparece o valor P(3) = 70

```
(Geral)                                         Horner_sub

Sub Horner_sub()


Dim x As Double, P As Double
Dim n As Integer
Dim a(10) As Double

x = InputBox("Qual é o valor de x?")
n = InputBox("Qual o grau do polinômio?")


For i = 0 To n Step 1
a(i) = InputBox("Introduza o coeficiente a(" & i & ")")
Next

'MsgBox "os coeficientes são: " & a(0) & "," & a(1) & "," & a(2) & "," & a(3) _
'& "," & a(4)
For i = 0 To n Step 1
    MsgBox "os coeficientes são: " & a(i)
Next
 P = 0
 For i = 0 To n Step 1

  P = P * x + a(i)

  Next
     MsgBox " P(" & x & ") = " & P

End Sub
```

Pode-se modificar esta macro para determinar um zero de P(x), ajustando o valor de x até que P(x) ≈ 0

O número PI

O valor de π é conhecido com extraordinária precisão há muito tempo. Já Arquimedes de Siracusa (cerca de 287-212 a.C.) estimou que o valor de π estava situado entre 3,14103 e 3,14271. Ele estudava polígonos regulares, de 96 lados, inscritos em um círculo e circunscrevendo um círculo.

Utilizando um polígono de 393216 lados, Vieta (1579) determinou o valor de π com 9 casas decimais.

Um quadrado, inscrito em um círculo de raio 1, tem lados de comprimento $s_4 = \sqrt{2}$, um polígono com 8 lados tem $s_{2n} = s_8 = \sqrt{2 - \sqrt{4 - s_4 s_4}} = \sqrt{2 - \sqrt{2}}$. Para um polígono de 16 lados resulta $s_{2n} = s_{16} = \sqrt{2 - \sqrt{4 - s_8 s_8}} = \sqrt{2 - \sqrt{2 + \sqrt{2}}}$.

Assim, podemos escrever em forma geral

$$s_{2n} = \sqrt{2 - \sqrt{4 - \sqrt{s_n s_n}}} \qquad (1)$$

Para desenvolver uma planilha para a equação (1), precisamos de três colunas, B para o números dos lados, C para os comprimentos e D para o valor aproximativo de π.

B5: 4; C5: =RAIZ(2); D5: =B5*C5/2
B6: =2*B5; C6: =RAIZ(2-RAIZ(4-C5*C5)); D6: =B6*C6/2.

Copiar B6:D6 até B6:D40

C6 =RAIZ(2-RAIZ(4-C5*C5))

	A	B	C	D	E
1					
2			**Arquimedes e Pi**		
3		Número de lados	Comprimento s	Aproximação para Pi	
4					
5		4	1,414213562373	2,828427124746	
6		8	0,765366864730	3,061467458921	
7		16	0,390180644032	3,121445152258	
8		32	0,196034280659	3,136548490546	
9		64	0,098135348655	3,140331156955	
10		128	0,049082457046	3,141277250933	
11		256	0,024543076571	3,141513801144	
12		512	0,012271769298	3,141572940368	
13		1024	0,006135913526	3,141587725280	
14		2048	0,003067960373	3,141591421505	
15		4096	0,001533980638	3,141592345611	
16		8192	0,000766990375	3,141592576545	
17		16384	0,000383495195	3,141592633463	
18		32768	0,000191747599	3,141592654808	

Tudo anda bem até n = 32768. Este polígono produz um valor para π correto com oito casas decimais.

25		4194304	0,000001497993	3,141518840466	
26		8388608	0,000000748922	3,141207968282	
27		16777216	0,000000374609	3,142451272494	
28		33554432	0,000000187305	3,142451272494	
29		67108864	0,000000094243	3,162277660168	
30		134217728	0,000000047122	3,162277660168	
31		268435456	0,000000021073	2,828427124746	
32		536870912	0,000000000000	0,000000000000	
33		1073741824	0,000000000000	0,000000000000	
34		2147483648	0,000000000000	0,000000000000	
35		4294967296	0,000000000000	0,000000000000	

Mas, depois disso, começa o caos e dá para n = 536870912 o valor π = 0.

Como podemos entender isso?

Deve-se saber que uma computador trabalha com números reais e que cada computador utiliza um número limitado de dígitos para representar um número real (uma "palavra"). Os computadores utilizam diferentes arquiteturas para representar uma palavra. Os micros utilizam palavras de comprimento 16 bits (dígitos binários), estações de trabalho usam uma palavra de 32 bits e os supercomputadores usam palavras de comprimento 64 bits.

Suponhamos, por simplicidade, que o nosso computador somente possa processar números com quatro dígitos. Um número como 3.141 será representado na forma padrão como .3141E-1. Nesta representação normal, uma palavra começa com um ponto digital e, detrás do ponto, não pode haver como primeiro digito um digito zero. O número 0.0069 será armazenado como .6900E-2. Observe que ganhamos dois zeros, dos quais não sabemos se são verdadeiros ou falsos.

Ao somar .3141E-1 e .6900E-2, devemos, primeiro, equalizar os expoentes:

```
   .3141E-1
+  .0690E-1
___________

=  .3831E-1
```

Como o acrescentar de zeros não-significantes pode conduzir a erros, podemos facilmente ver se repetimos o cálculo de π com só 3 casas decimais

```
S4 = .141E1
S8 = RAIZ(.200E1-.142E1)=.762E+0
S16= RAIZ(.200E1-.185E1)=.387E+0
S32= RAIZ(.200E1-.196E1)=.200E+0
S64= RAIZ(.200E1-.199E1)=.100E+0
S128= RAIZ(.200E1-.200E1)= 0E+0
```

Compare S16:

RAIZ(.200E1-.185E1) = RAIZ(.015E1) = RAIZ(.15**0**E+0)= .387E+0

O zero marcado foi adicionado pela normalização. Estes zeros extras fazem com que a partir de S128 resultem só zeros.

Como podemos evitar este dilema? O problema principal são as duas subtrações na fórmula recursiva (1). Devemos tratar de transformar uma das subtrações numa adição, coisa que não é difícil, pois é só multiplicar $s = \sqrt{2-\sqrt{4-s\cdot s}}$ por $\frac{\sqrt{2+\sqrt{4-s\cdot s}}}{\sqrt{2+\sqrt{4-s\cdot s}}}$. Obteremos, assim, a seguinte formula

$$s_{2n} = \frac{s_n}{\sqrt{2+\sqrt{4-s_n s_n}}} \qquad (2)$$

Da planilha a seguir podemos ver como a nova fórmula da valores estáveis e corretos até 14 casas decimais a partir de n = 33554432

17		16384	0,000383495195	3,14159263433856
18		32768	0,000191747598	3,14159264877699
19		65536	0,000095873799	3,14159265238659
20		131072	0,000047936900	3,14159265328899
21		262144	0,000023968450	3,14159265351459
22		524288	0,000011984225	3,14159265357099
23		1048576	0,000005992112	3,14159265358509
24		2097152	0,000002996056	3,14159265358862
25		4194304	0,000001498028	3,14159265358950
26		8388608	0,000000749014	3,14159265358972
27		16777216	0,000000374507	3,14159265358978
28		33554432	0,000000187254	3,14159265358979
29		67108864	0,000000093627	3,14159265358979
30		134217728	0,000000046813	3,14159265358979
31		268435456	0,000000023407	3,14159265358979
32		536870912	0,000000011703	3,14159265358979
33		1073741824	0,000000005852	3,14159265358979
34		2147483648	0,000000002926	3,14159265358979
35		4294967296	0,000000001463	3,14159265358979

A macro "ArchiPi" utiliza `DIM n As Long` , para poder ter uma exatidão de 1E-14.

```
Sub ArchiPi()

Dim n As Long, s As Double, pi As Double
    Dim t As Double, y As Double

    n = 4: s = 2 ^ 0.5: pi = 3
    Do While Abs(pi - t) >= 0.00000000000001 '=1E-14
    t = pi 'valor velho de pi
    s = s / (2 + (4 - s * s) ^ 0.5) ^ 0.5 '(2 - (4 - s * s) ^ 0.5) ^ 0.5
    pi = s * n 'novo valor de pi
    n = 2 * n

    Loop

    MsgBox "n= " & n & ", Pi= " & pi

End Sub
```



Dos algoritmos, baseando-se numa série de potências, o mais conhecido é nomeado segundo Leibniz (1646-1716, político, matemático, inventor do cálculo, físico, diplomata, jurista, filósofo...)

$$\pi = 4(1-\frac{1}{3}+\frac{1}{5}-\frac{1}{7}+...) = 4\sum_{n=1}^{\infty}\frac{(-1)^{n-1}}{2n-1} \quad (3)$$

Lamentavelmente, a velocidade de convergência da serie (3) é muito lenta.

```
(Geral)                                   Leibniz

Sub Leibniz()
    Dim n As Integer
    Dim x As Double, y As Double, s As Double

    y = 1: n = 3: x = -1 / n
    Do While Abs(x) > 0.0001
    y = y + x
    s = 4 * y
    x = -x * n / (n + 2)
    n = n + 2
    Loop
    MsgBox "A soma é s = " & s


End Sub
```

Agora, quero demonstrar lhes um excelente algoritmo nomeado segundo J. Gregory (1671) e J. Machin (1706).

$$\frac{\pi}{4} = 4\sum_{n=1}^{\infty} A_n - \sum_{n=1}^{\infty} B_n \quad (4)$$

Os coeficientes são $A_n = \frac{(-1)^{n-1}}{(2n-1)5^{2n-1}}, \quad B_n = \frac{(-1)^{n-1}}{(2n-1)239^{2n-1}}$

A implementação em Excel pode dar a impressão de ser complicada, mas não é tão difícil assim. Vejamos:

A10: 3; B10: -1; C9: 5; C10 =25*C9

D9: =1/C9;   D10: =D9+B10/(A10*C10)
A11: =A10+2; B11: =-B10; C11: =25*C10

D11: =D10+B11/(A11*C11)

Copiar A11:D11 até A18:D18

E18: =16*D18; C19: 239 (começa o cálculo dos Bn); D19: =1/C19
A20: 3; B20: -1; C20: =57121*C19

D20: =D19+B20/(A20*C20)
A21: =A20+2; B21: =-B20; C21: =57121*C20
D21: =D20+B21/(A21*C21); copiar A21:D21 até A22:D22
E22: =4*D22
E6: =E18-E22 (aproximação para Pi)

E6 =E18-E22

	A	B	C	D	E	F	G	H
1								
2	**Pi segundo GREGORY-MACHIN**							
3								
4								
5								
6				**Pi=**	3,141592653589790		(14 casas decimais)	
7								
8								
9			5	0,2				
10	3	-1	125	0,1973333				
11	5	1	3125	0,1973973				
12	7	-1	78125	0,1973955				
13	9	1	1953125	0,1973956				
14	11	-1	48828125	0,1973956				
15	13	1	1220703125	0,1973956				
16	15	-1	30517578125	0,1973956				
17	17	1	7,62939E+11	0,1973956				
18	19	-1	1,90735E+13	0,1973956	3,1583289575981			
19			239	0,00418410				
20	3	-1	13651919	0,0041841				
21	5	1	7,79811E+11	0,0041841				
22	7	-1	4,45436E+16	0,0041841	0,0167363040083			
23								

Se precisa somente 9 termos da série An e 3 da série Bn para obter π com 13 casas decimais.
A fórmula por detrás do algoritmo é Pi = 16 arctan(1/5) - 4 arctan(1/239)

O algoritmo-Pi dos irmãos Borwein

Em 1995, Jonathan Borwein e Peter Borwein descobriram um algoritmo para determinar o valor de 1/π com um extraordinário desempenho. Ele fornece π com 170 casas decimais corretas após apenas 3 passos de iteração.
Em 1995, os irmãos Borwein determinaram, em colaboração com Yasumasa Kanada da Universidade de Tóquio, o número π com 6,4 bilhões de casas decimais, batendo o recorde mundial no calculo de π.

O algoritmo reza assim:

$$y_0 = \sqrt{2} - 1$$

$$a_0 = 6 - 4\sqrt{2}$$

$$y_{k+1} = \frac{1 - \sqrt[4]{1 - y_k^4}}{1 + \sqrt[4]{1 - y_k^4}}$$

$$a_{k+1} = a_k (1 + y_{k+1})^4 - 2^{2k+3} y_{k+1} (1 + y_{k+1} + y_{k+1}^2)$$

O algoritmo tem um aspecto assustador, mas, como veremos, ele é fácil a implementar. É importante reduzir tudo a operações elementares, ou seja, as potencias e raízes devem ser reduzidas a operações de adição e de multiplicação. Mas, mesmo calculando √2 como potencia $2^{1/2}$, a convergência é surpreendentemente rápida. A potência 2^{2k+3} pode ser calculada iterativamente como pot2 = pot2·4 com o valor inicial de pot2 = 2.
Vem aqui uma implementação do algoritmo:

	A	B
1	Pi	n
2	2,914213562373100	0
3	3,141592646213550	1
4	3,141592653589790	2
5	3,141592653589790	3
6	3,141592653589790	4
7		
8		
9		
10		
11		
12		
13		
14		
15		
16		

(Geral) Borwein_Pi

```
Sub Borwein_Pi()
 y = (2) ^ (1 / 2) - 1
 a = 6 - 4 * (2) ^ (1 / 2)
 pot2 = 2
 n = 3
 Cells(2, 1).Value = 1 / a
 Cells(2, 2).Value = 0
 For k = 0 To n Step 1
 r = ((1 - y * y * y * y) ^ (1 / 2)) ^ (1 / 2)
 y = (1 - r) / (1 + r)
 s = 1 + y
 s = s * s * s * s
 pot2 = pot2 * 4
 a = a * s - pot2 * y * (1 + y + y * y)
 Cells(3 + k, 1).Value = 1 / a
 Cells(3 + k, 2).Value = k + 1
 Next
 'MsgBox "Pi= " & Format(1 / ap, "0.000000000000000")
End Sub
```

SOMASEQÜÊNCIA

No Excel encontramos a função SOMASEQÜÊNCIA que retorna a soma de um polinômio e é baseada na expressão

$$\text{SOMASEQÜÊNCIA}(x,n,m,a) = a_1 x^n + a_2 x^{(n+m)} + a_3 x^{(n+2m)} + \ldots + a_i x^{(n+(i-1)m)}$$

Já sabemos que muitas funções podem ser aproximadas por um polinômio, ou seja por uma soma parcial de uma série de potências.

Sintaxe

SOMASEQÜÊNCIA(x;n;m;coeficientes)

x é o valor de entrada.
n é a potência inicial à qual você deseja elevar x.
m é o passo pelo qual se acrescenta n a cada termo na seqüência.
coeficientes são os fatores a_i dos termos do polinômio.

Exemplo:

A série da função cos é dada por $\cos(x) = 1 - \frac{x^2}{2!} + \frac{x^4}{4!} - \ldots$

Se queremos calcular cos(2), escrevemos x = 2. A potência inicial é n = 0, pois $x^0 = 1$. Os passos são m = 2, e os coeficientes são os fatores dos termos do polinômio, ou seja: 1/0!, -1/2!, 1/4!, -1/6! etc. (0! = 1 por definição.) Veja a seguinte planilha:

C5 | f_x =B5/FATORIAL(A5)

	A	B	C	D	E	F	G	H
1		**Coseno**						
2						x(radianos):	2	
3	n	1	1					
4	2	-1	-0,5					
5	4	1	0,041667					
6	6	-1	-0,00139			SOMASEQÜÊNCIA:		
7	8	1	2,48E-05		cos(x)=	-0,41614684		
8	10	-1	-2,8E-07					
9	12	1	2,09E-09					
10	14	-1	-1,1E-11					
11								

A fórmula em F7 é =SOMASEQÜÊNCIA(G2;0;2;C3:C10). Os coeficientes na coluna C são C3: 1; C4: =B4/FATORIAL(A4) etc.

Construímos a planilha para a função **Seno** da mesma forma:

C4 | f_x =B4/FATORIAL(A4)

	A	B	C	D	E	F	G	H
1		**Seno**						
2					x(radianos):		2	
3	1	1	1					
4	3	-1	-0,16667					
5	5	1	0,008333					
6	7	-1	-0,0002			SOMASEQÜÊNCIA:		
7	9	1	2,76E-06		sen(x)=	0,909297		
8	11	-1	-2,5E-08					
9	13	1	1,61E-10					
10	15	-1	-7,6E-13					
11								
12								

Agora, a fórmula em F7 é =SOMASEQÜÊNCIA(G2;1;2;C3:C10), pois a série da função seno é dada por

$$sen(x) = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + - ...$$

Os valores de n na coluna A foram devidamente selecionados.

Mas, parece que a função SOMASEQÜÊNCIA não é de grande utilidade, pois não faz outra coisa que somar os termos da série, veja a próxima planilha, onde adicionamos na coluna D as potências de *x* e utilizamos em F10: =SOMA(D3:D10) –os resultados são idênticos.

D3 | f_x =C3*G$2^A3

	A	B	C	D	E	F	G	H
1		**Seno**						
2						x(radianos):	2	
3	1	1	1	2				
4	3	-1	-0,16667	-1,33333				
5	5	1	0,008333	0,266667				
6	7	-1	-0,0002	-0,0254		SOMASEQÜÊNCIA:		
7	9	1	2,76E-06	0,001411	sen(x)=	0,909297		
8	11	-1	-2,5E-08	-5,1E-05				
9	13	1	1,61E-10	1,32E-06				
10	15	-1	-7,6E-13	-2,5E-08	Soma=	0,909297		
11								